Calculation of the charges of Kerr-Newman BH by the solution phase space method

Written by Kamal Hajian
Institute for Research in Fundamental Sciences (IPM), School of Physics
Based on the paper: K. Hajian and M.M. Sheikh-Jabbari, Phys.Rev. D93 (2016) no.4, 044074 , arXiv:1512.05584.

IPM Second Workshop on xAct, October 19-21, 2016.

The Kerr-Newman solution

Introducing the manifold and the chart

In[1]:=
```
Block[{Print}, << xAct`xTras`]; Block[{Print}, << xAct`TexAct`];
$DefInfoQ = False;
$UndefInfoQ = False;
$CVVerbose = False;
$CovDFormat = "Prefix";
$CommuteCovDsOnScalars = True;


DefManifold[M, 4, {α, β, γ, δ, ξ, ι, κ, λ, μ, ο, ς, ρ, σ, τ, υ, ω, ν}]
DefMetric[-1, g[-α, -β], CD, PrintAs → "g"]
DefChart[B, M, {0, 1, 2, 3}, {t[], r[], θ[], φ[]}, ChartColor → Blue]
```

This is a command for simplifying any result defined on the chart. Hence, by this command, one can find the explicit components of a given tensor calculated on the background metric. The command is "MySimplify[]"

In[10]:=
```
MySimplify1[a_] := ChangeCovD[a, CD, PDB];
MySimplify2[b_] := ToBasis[B]@ToBasis[B]@MySimplify1[b];
MySimplify3[c_] := TraceBasisDummy@MySimplify2[c];
MySimplify4[d_] := ComponentArray@MySimplify3[d];
MySimplify5[e_] := Factor@ToValues@ToValues@ToValues@MySimplify4[e];
MySimplify[f_] := Simplify[MySimplify5[f], TimeConstraint → 1000]
```

Defining the Kerr-Newman metric:

In[16]:=
```
DefConstantSymbol[G, PrintAs → "G"]
DefConstantSymbol[m, PrintAs → "m"]
DefConstantSymbol[a, PrintAs → "a"]
DefConstantSymbol[q, PrintAs → "q"]
$Assumptions = And[r[] ∈ Reals, r[] > 0, θ[] ∈ Reals, 0 < θ[] < π, G ∈ Reals,
   G > 0, m ∈ Reals, m >= 0, a ∈ Reals, a >= 0, q ∈ Reals, q >= 0, G^2 m^2 ≥ a^2 + q^2];
```

The Kerr-Newman metric

In[21]:=
```
ρ2 = r[]^2 + a^2 Cos[θ[]]^2;
Δ = r[]^2 - 2 G m r[] + a^2 + q^2;
f = (2 G m r[] - q^2)/ρ2;
```

In[24]:=
```
MatrixForm[TheMetric = {{-(1 - f), 0, 0, -f a Sin[θ[]]^2}, {0, ρ2/Δ, 0, 0},
     {0, 0, ρ2, 0},
     {-f a Sin[θ[]]^2, 0, 0, (r[]^2 + a^2 + f a^2 Sin[θ[]]^2) Sin[θ[]]^2}}]
```

Out[24]//MatrixForm=

$$\begin{pmatrix} -1+\frac{-q^2+2Gmr}{a^2\cos[\theta]^2+r^2} & 0 & 0 & -\frac{a(-q^2+2Gmr)\sin[\theta]^2}{a^2\cos[\theta]^2+r^2} \\ 0 & \frac{a^2\cos[\theta]^2+r^2}{a^2+q^2-2Gmr+r^2} & 0 & 0 \\ 0 & 0 & a^2\cos[\theta]^2+r^2 & 0 \\ -\frac{a(-q^2+2Gmr)\sin[\theta]^2}{a^2\cos[\theta]^2+r^2} & 0 & 0 & \sin[\theta]^2\left(a^2+r^2+\frac{a^2(-q^2+2Gmr)\sin[\theta]^2}{a^2\cos[\theta]^2+r^2}\right) \end{pmatrix}$$

In[25]:=
```
MetricInBasis[g, -B, TheMetric];
```

In[26]:=
```
$CommutePDs = True;
$CommutePDBs = True;
```

The Kerr-Newman gauge field

In[28]:=
```
DefTensor[A[-α], M]
DefTensor[F[-α, -β], M, Antisymmetric[{-α, -β}]]
```

In[30]:=
```
RuleF = MakeRule[{F[-α, -β], CD[-α]@A[-β] - CD[-β]@A[-α]}, MetricOn → All]
```

Out[30]= {HoldPattern[$F^{\underline{\alpha}\underline{\beta}}$] :→ Module[{}, $\nabla^{\alpha} A^{\beta} - \nabla^{\beta} A^{\alpha}$]}

In[31]:=
```
AllComponentValues[A[{-α, -B}], {q r[]/ρ2, 0, 0, -q a r[] Sin[θ[]]^2/ρ2}];
ChangeComponents[A[{α, B}], A[{-ρ, -B}]];
```

Computed $A^{\alpha} \to A_{\beta}\ g^{\alpha\beta}$ in 0.042587 Seconds

Calculating curvature tensors etc:

In[33]:=
```
MetricCompute[g, B, All]
```

In[34]:=
```
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {σ, B}, {-τ, -B}],
  MySimplify[g[σ, o] Christoffel[CD, PDB][ρ, -o, -τ]]];
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {σ, B}, {τ, B}],
  MySimplify[g[τ, α] g[σ, o] Christoffel[CD, PDB][ρ, -o, -α]]];
AllComponentValues[Christoffel[CD, PDB][{ρ, B}, {-σ, -B}, {τ, B}],
  MySimplify[g[τ, α] Christoffel[CD, PDB][ρ, -σ, -α]]];
AllComponentValues[Christoffel[CD, PDB][{-ρ, -B}, {-σ, -B}, {τ, B}],
  MySimplify[g[-ρ, -β] g[τ, α] Christoffel[CD, PDB][β, -σ, -α]]];
```

In[38]:=
```
ChangeComponents[RicciCD[{α, B}, {β, B}], RicciCD[{-α, -B}, {-β, -B}]];
```

Computed $R[\nabla]_{\alpha}{}^{\beta} \to g^{\beta\gamma}\, R[\nabla]_{\alpha\gamma}$ in 0.383814 Seconds

Computed $R[\nabla]^{\alpha\beta} \to g^{\alpha\gamma}\, R[\nabla]_{\gamma}{}^{\beta}$ in 0.378811 Seconds

e.o.m

Here, we find the equations of motion through
$\delta\left(\sqrt{-g}\ \mathcal{L}\right) = \sqrt{-g}\left[(\mathrm{eom}_g)^{\alpha\beta}\ \delta g_{\alpha\beta} + (\mathrm{eom}_A)^{\alpha}\ \delta A_{\alpha}\right] + a$ surface term

In[39]:=
```
ℒ = 1/(16 π G) (RicciScalarCD[] - F[-α, -β] F[α, β]) /. RuleF;
(VarL[g[-α, -β]][ℒ]) // ToCanonical // ContractMetric // Simplify //
   ContractMetric // ToCanonical // Factor
```

Out[40]=
$$-\frac{1}{32\,G\,\pi}\Big(2\,R[\nabla]^{\alpha\beta} - g^{\alpha\beta}\,R[\nabla] - 4\left(\nabla^{\alpha}A^{\gamma}\right)\left(\nabla^{\beta}A_{\gamma}\right) + 4\left(\nabla^{\beta}A_{\gamma}\right)\left(\nabla^{\gamma}A^{\alpha}\right) - 4\left(\nabla_{\gamma}A^{\beta}\right)\left(\nabla^{\gamma}A^{\alpha}\right) + 4\left(\nabla^{\alpha}A_{\gamma}\right)\left(\nabla^{\gamma}A^{\beta}\right) - 2\,g^{\alpha\beta}\left(\nabla_{\gamma}A_{\delta}\right)\left(\nabla^{\delta}A^{\gamma}\right) + 2\,g^{\alpha\beta}\left(\nabla_{\delta}A_{\gamma}\right)\left(\nabla^{\delta}A^{\gamma}\right)\Big)$$

In[41]:=
```
eomg = % * g[-α, -μ] g[-β, -ν] // ContractMetric // ToCanonical // Simplify
```

Out[41]=
$$\frac{1}{32\,G\,\pi}\Big(-2\,R[\nabla]_{\mu\nu} + g_{\mu\nu}\left(R[\nabla] + 2\left(\nabla_{\alpha}A_{\beta}\right)\left(\nabla^{\beta}A^{\alpha}\right) - 2\left(\nabla_{\beta}A_{\alpha}\right)\left(\nabla^{\beta}A^{\alpha}\right)\right) + 4\left(\left(\nabla_{\alpha}A_{\nu}\right)\left(\nabla^{\alpha}A_{\mu}\right) - \left(\nabla^{\alpha}A_{\nu}\right)\left(\nabla_{\mu}A_{\alpha}\right) - \left(\nabla^{\alpha}A_{\mu}\right)\left(\nabla_{\nu}A_{\alpha}\right) + \left(\nabla_{\mu}A^{\alpha}\right)\left(\nabla_{\nu}A_{\alpha}\right)\right)\Big)$$

In[42]:=
```
eomA =
  (VarD[A[-α], CD][ℒ]) // ToCanonical // ContractMetric // ToCanonical // Simplify
```

Out[42]=
$$\frac{-\left(\nabla_{\beta}\nabla^{\alpha}A^{\beta}\right) + \nabla_{\beta}\nabla^{\beta}A^{\alpha}}{4\,G\,\pi}$$

We can check that the Kerr-Newman geometry satisfies these equations

In[43]:=
```
eomg // ToBasis[B] // ToBasis[B] // TraceBasisDummy // ComponentArray // MySimplify
```

Out[43]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[44]:=
```
eomA // MySimplify
```

Out[44]= {0, 0, 0, 0}

Now, we define rules to impose the on-shell condition wherever we want

In[45]:=
```
Ruleeomg = MakeRule[
  Evaluate[{RicciCD[-μ, -ν], (16 π G * (eomg) + RicciCD[-μ, -ν]) // Simplify}],
  MetricOn → All]
```

Out[45]= $\{\text{HoldPattern}[R[\nabla]^{\underline{\mu}\underline{\nu}}] :\to \text{Module}[\{\alpha, \beta\}, \frac{1}{2} g^{\nu\mu} R[\nabla] + 2 (\nabla^{\mu} A^{\alpha}) (\nabla^{\nu} A_{\alpha}) - 2 (\nabla^{\nu} A_{\alpha}) (\nabla^{\alpha} A^{\mu}) + 2 (\nabla_{\alpha} A^{\nu}) (\nabla^{\alpha} A^{\mu}) - 2 (\nabla^{\mu} A_{\alpha}) (\nabla^{\alpha} A^{\nu}) + g^{\nu\mu} (\nabla_{\alpha} A_{\beta}) (\nabla^{\beta} A^{\alpha}) - g^{\nu\mu} (\nabla_{\beta} A_{\alpha}) (\nabla^{\beta} A^{\alpha})]\}$

In[46]:=
```
RuleeomA =
 MakeRule[Evaluate[{CD[-β][CD[β][A[α]]], (-4 π G (eomA) + CD[-β][CD[β][A[α]]]) //
     FullSimplification[]}], MetricOn → All]
```

Out[46]= $\{\text{HoldPattern}[\nabla_{\underline{\beta}} \nabla^{\underline{\beta}} A^{\underline{\alpha}}] :\to \text{Module}[\{\gamma\}, \nabla_{\gamma} \nabla^{\alpha} A^{\gamma}], \text{HoldPattern}[\nabla^{\underline{\beta}} \nabla_{\underline{\beta}} A^{\underline{\alpha}}] :\to \text{Module}[\{\gamma\}, \nabla_{\gamma} \nabla^{\alpha} A^{\gamma}]\}$

To cross check, we can check the vanishing of the equations of motion via on-shell rules defined above:

In[47]:=
```
eomg /. Ruleeomg // ToCanonical // FullSimplification[]
```

Out[47]= 0

In[48]:=
```
eomA /. RuleeomA // ToCanonical // FullSimplification[]
```

Out[48]= 0

Charge Calculation

Finding the Θ^{μ} surface term for the Einstein-Maxwell theory:

We can find the surface term Θ^{μ} by variating the Lagrangian with respect to the dynamical field
$\delta\left(\sqrt{-g}\,\mathcal{L}\right) = \sqrt{-g}\left[(\text{eom}_g)^{\alpha\beta}\,\delta g_{\alpha\beta} + (\text{eom}_A)^{\alpha}\,\delta A_{\alpha}\right] + \sqrt{-g}\,\nabla_{\mu}\Theta^{\mu}$

In[49]:=
```
ExpandPerturbation@Perturbation[Sqrt[-Detg[]] ℒ] // ContractMetric //
  ToCanonical // Factor
```

Out[49]= $-\frac{1}{32 G \pi}\sqrt{-\tilde{g}}\,\big(2\,\Delta g^{1\alpha\beta}\,R[\nabla]_{\alpha\beta} - \Delta g^{1\alpha}{}_{\alpha}\,R[\nabla] - 2 (\nabla_{\beta}\nabla_{\alpha}\Delta g^{1\alpha\beta}) + 2 (\nabla_{\beta}\nabla^{\beta}\Delta g^{1\alpha}{}_{\alpha}) - 2\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla_{\alpha}A_{\beta})(\nabla^{\beta}A^{\alpha}) - 8 (\nabla_{\alpha}\Delta[A_{\beta}])(\nabla^{\beta}A^{\alpha}) + 2\,\Delta g^{1\gamma}{}_{\gamma}\,(\nabla_{\beta}A_{\alpha})(\nabla^{\beta}A^{\alpha}) - 4\,\Delta g^{1}{}_{\alpha\gamma}\,(\nabla_{\beta}A^{\gamma})(\nabla^{\beta}A^{\alpha}) + 8 (\nabla_{\beta}\Delta[A_{\alpha}])(\nabla^{\beta}A^{\alpha}) - 4\,\Delta g^{1}{}_{\beta\gamma}\,(\nabla^{\beta}A^{\alpha})(\nabla^{\gamma}A_{\alpha}) + 8\,\Delta g^{1}{}_{\alpha\gamma}\,(\nabla^{\beta}A^{\alpha})(\nabla^{\gamma}A_{\beta})\big)$

Subtracting the equation of motion, the result is a total derivative

In[50]:=
```
% - Sqrt[-Detg[]] eomg * Perturbationg[LI[1], μ, ν] -
    Sqrt[-Detg[]] eomA * Perturbation[A[-α]] // ContractMetric //
  ToCanonical // Factor // Simplify // FullSimplify
```

Out[50]= $\frac{1}{16\, G\, \pi}\sqrt{-\tilde{\tilde{g}}}\,\left(\nabla_\beta \nabla_\alpha \Delta g^{1\alpha\beta} + 4\, \Delta[A^\alpha]\,\left(\nabla_\beta \nabla_\alpha A^\beta - \nabla_\beta \nabla^\beta A_\alpha\right) - \nabla_\beta \nabla^\beta \Delta g^{1\alpha}{}_\alpha + 4\,\left(\nabla_\alpha \Delta[A_\beta] - \nabla_\beta \Delta[A_\alpha]\right)\,\left(\nabla^\beta A^\alpha\right)\right)$

To make sure, subtracting the following total derivative term, would result to zero:

In[51]:=
```
(% - (Sqrt[-Detg[]] * CD[-β] @ (CD[-α][Perturbationg[LI[1], α, β]] +
   4 * Perturbation[A[α]] * (CD[-α][A[β]] - CD[β][A[-α]]) -
  CD[β][Perturbationg[LI[1], α, -α]])) /
 (16 * G * Pi)) // FullSimplification[]
```

Out[51]= 0

Dropping the divergence, the rest would be the Θ^μ. So, we make a rule to identify this tensor.

In[52]:=
```
DefTensor[Θ[μ], M]
RuleΘ = MakeRule[{Θ[β], ((CD[-α][Perturbationg[LI[1], α, β]] +
   4 * Perturbation[A[α]] * (CD[-α][A[β]] - CD[β][A[-α]]) -
  CD[β][Perturbationg[LI[1], α, -α]])) /
 (16 * G * Pi)}, MetricOn → All]
```

Out[53]= $\{\text{HoldPattern}[\Theta^{\underline{\beta}}] \mapsto \text{Module}\left[\{\alpha\},\ -\frac{\Delta[A^\alpha]\,(\nabla^\beta A_\alpha)}{4\, G\, \pi} - \frac{\nabla^\beta \Delta g^{1\alpha}{}_\alpha}{16\, G\, \pi} + \frac{\Delta[A^\alpha]\,(\nabla_\alpha A^\beta)}{4\, G\, \pi} + \frac{\nabla_\alpha \Delta g^{1\alpha\beta}}{16\, G\, \pi}\right]\}$

Finding the Noether charge density $Q^{\mu\nu}$ for the E-M theory and the generator ϵ:

In order to find the Noether charge $Q^{\mu\nu}$ associated with the generator $\epsilon = \{\xi^\mu, \lambda\}$, first we need to find the Noether current $J_\epsilon{}^\mu = \Theta^\mu\,(\delta_\epsilon g,\ \delta_\epsilon A) - \xi^\mu\,\mathcal{L}$. Then, using the on-shell relation $J_\epsilon{}^\mu = \nabla_\nu Q^{\mu\nu}$ we can find the $Q^{\mu\nu}$

The vector ξ^μ acts on the fields as Lie variaiton. Besides, under the gauge transformation we have $A_\mu \to A_\mu + \partial_\mu \lambda$. Here, we provide rules to imply it

In[54]:=
```
DefTensor[ξ[μ], M]
DefTensor[lambda[], M, PrintAs → "λ"]
```

In[56]:=
```
Ruledeleg = MakeRule[Evaluate[{Perturbationg[LI[1], -α, -β],
   LieD[ξ[μ], CD][g[-α, -β]] // ContractMetric // ToCanonical}], MetricOn → All]
```

Out[56]= $\{\text{HoldPattern}[\Delta g^{1\underline{\alpha}\underline{\beta}}] \mapsto \text{Module}[\{\},\ \nabla^\alpha \xi^\beta + \nabla^\beta \xi^\alpha]\}$

In[57]:=
```
RuledeleA = MakeRule[Evaluate[{Perturbation[A[-α]],
     (LieD[ξ[μ], CD][A[-α]] // ContractMetric // ToCanonical) +
      CD[-α][lambda[]]}], MetricOn → All]
```

Out[57]= $\{\text{HoldPattern}[\Delta[A^{\underline{\alpha}}]] \mapsto \text{Module}[\{\beta\}, \nabla^{\alpha}\lambda + A^{\beta}(\nabla^{\alpha}\xi_{\beta}) + \xi^{\beta}(\nabla_{\beta}A^{\alpha})]\}$

Finding the J_{ϵ}^{μ}

In[58]:=
```
(Θ[μ] /. RuleΘ /. Ruledeleg /. RuledeleA) - ξ[μ] ℒ // ContractMetric // ToCanonical //
 Factor
```

Out[58]= $-\frac{1}{16\,G\,\pi}\big(R[\nabla]\,\xi^{\mu} - \nabla_{\alpha}\nabla^{\alpha}\xi^{\mu} - \nabla_{\alpha}\nabla^{\mu}\xi^{\alpha} - 4(\nabla_{\alpha}\lambda)(\nabla^{\alpha}A^{\mu}) + 2\,\xi^{\mu}(\nabla_{\alpha}A_{\beta})(\nabla^{\beta}A^{\alpha}) - 2\,\xi^{\mu}(\nabla_{\beta}A_{\alpha})(\nabla^{\beta}A^{\alpha}) - 4\,\xi^{\alpha}(\nabla_{\alpha}A_{\beta})(\nabla^{\beta}A^{\mu}) - 4\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha})(\nabla^{\beta}A^{\mu}) + 4(\nabla_{\alpha}\lambda)(\nabla^{\mu}A^{\alpha}) + 4\,\xi^{\alpha}(\nabla_{\alpha}A_{\beta})(\nabla^{\mu}A^{\beta}) + 4\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha})(\nabla^{\mu}A^{\beta}) + 2(\nabla^{\mu}\nabla_{\alpha}\xi^{\alpha})\big)$

J_{ϵ}^{μ} is a divergence term on-shell.

In[59]:=
```
Jϵμ = (% // Expand // FullSimplification[]) /. Ruleeomg /. RuleeomA // Factor //
   ToCanonical // FullSimplify
```

Out[59]= $\frac{1}{32\,G\,\pi}\big(-\xi^{\mu}(R[\nabla] + 2(\nabla_{\alpha}A_{\beta} - \nabla_{\beta}A_{\alpha})(\nabla^{\beta}A^{\alpha})) + 2(\nabla_{\alpha}\nabla^{\alpha}\xi^{\mu} + 4(\nabla_{\alpha}\lambda)(\nabla^{\alpha}A^{\mu} - \nabla^{\mu}A^{\alpha}) + 2(\xi^{\alpha}(\nabla_{\alpha}A_{\beta} + \nabla_{\beta}A_{\alpha}) + 2\,A^{\alpha}(\nabla_{\beta}\xi_{\alpha}))(\nabla^{\beta}A^{\mu} - \nabla^{\mu}A^{\beta}) - \nabla^{\mu}\nabla_{\alpha}\xi^{\alpha})\big)$

Dropping the divergence, we define the Noether charge density $Q_{\epsilon}^{\mu\nu}$

In[60]:=
```
DefTensor[Q[μ, ν], M]
RuleQ =
 MakeRule[{Q[μ, ν], (CD[ν][ξ[μ]] - CD[μ][ξ[ν]] - 4 (CD[μ]@A[ν] - CD[ν]@A[μ])
        (A[β] * ξ[-β] + lambda[])) / (16 π G)}, MetricOn → All]
```

Out[61]= $\{\text{HoldPattern}[Q^{\underline{\mu\nu}}] \mapsto \text{Module}[\{\alpha\}, -\frac{\lambda(\nabla^{\mu}A^{\nu})}{4\,G\,\pi} - \frac{A^{\alpha}\,\xi_{\alpha}(\nabla^{\mu}A^{\nu})}{4\,G\,\pi} - \frac{\nabla^{\mu}\xi^{\nu}}{16\,G\,\pi} + \frac{\lambda(\nabla^{\nu}A^{\mu})}{4\,G\,\pi} + \frac{A^{\alpha}\,\xi_{\alpha}(\nabla^{\nu}A^{\mu})}{4\,G\,\pi} + \frac{\nabla^{\nu}\xi^{\mu}}{16\,G\,\pi}]\}$

To cross check, let's check the vanishing of the J_{ϵ}^{μ} - $\nabla_{\nu}Q_{\epsilon}^{\mu\nu}$ on-shell:

In[62]:=
```
((Jϵμ - CD[-ν]@Q[μ, ν]) /. RuleQ /. Ruleeomg /. RuleeomA // FullSimplification[] //
     ContractMetric) /. Ruleeomg /. RuleeomA // ToCanonical // FullSimplify
```

Out[62]= 0

Finding the important 2-form $k^{\mu\nu}$ for the E-M theory and the generator ϵ:

Finding the most important tensor for charge calculations using $\sqrt{-g}\; k_{\epsilon}^{\mu\nu} = \delta(\sqrt{-g}\; Q_{\epsilon}^{\mu\nu})$ - $\sqrt{-g}\;(\xi^{\nu}\,\Theta^{\mu}$-$\xi^{\mu}\,\Theta^{\nu})$

In order to find the $\delta(\sqrt{-g}\ Q_\epsilon{}^{\mu\nu})$, we need rules preventing the δ to act on the ϵ

In[63]:=
```
RuleΔξ = MakeRule[{Perturbation[ξ[μ]], 0}, MetricOn → All]
RuleΔλ = MakeRule[{Perturbation[lambda[]], 0}]
```

Out[63]= $\{\text{HoldPattern}[\Delta[\xi^{\underline{\mu}}]] \mapsto \text{Module}[\{\}, 0]\}$

Out[64]= $\{\text{HoldPattern}[\Delta[\lambda]] \mapsto \text{Module}[\{\}, 0]\}$

Finding the $\delta(\sqrt{-g}\ Q_\epsilon{}^{\mu\nu})$:

In[65]:=
```
(Sqrt[-Detg[]] * Q[α, β]) /. RuleQ;
δQϵ = (ExpandPerturbation@Perturbation[%] // ContractMetric // ToCanonical //
     Factor) /. RuleΔξ /. RuleΔλ
```

Out[66]=
$$\begin{aligned}
-\frac{1}{32\,G\,\pi}\sqrt{-\tilde{\tilde{g}}}\ \big(&4\,\lambda\,\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\alpha A^\beta) + 4\,A^\gamma\,\Delta g1^{\delta}{}_{\delta}\,\xi_\gamma\,(\nabla^\alpha A^\beta) + 8\,\Delta[A_\gamma]\,\xi^\gamma\,(\nabla^\alpha A^\beta) - \\
&8\,\lambda\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\alpha A^\gamma) - 8\,A^\gamma\,\Delta g1^{\beta}{}_{\delta}\,\xi_\gamma\,(\nabla^\alpha A^\delta) + 8\,\lambda\,(\nabla^\alpha \Delta[A^\beta]) + \\
&8\,A^\gamma\,\xi_\gamma\,(\nabla^\alpha \Delta[A^\beta]) + 2\,\xi^\gamma\,(\nabla^\alpha \Delta g1^{\beta}{}_{\gamma}) + \Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\alpha \xi^\beta) - 4\,\lambda\,\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\beta A^\alpha) - \\
&4\,A^\gamma\,\Delta g1^{\delta}{}_{\delta}\,\xi_\gamma\,(\nabla^\beta A^\alpha) - 8\,\Delta[A_\gamma]\,\xi^\gamma\,(\nabla^\beta A^\alpha) + 8\,\lambda\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\beta A^\gamma) + \\
&8\,A^\gamma\,\Delta g1^{\alpha}{}_{\delta}\,\xi_\gamma\,(\nabla^\beta A^\delta) - 8\,\lambda\,(\nabla^\beta \Delta[A^\alpha]) - 8\,A^\gamma\,\xi_\gamma\,(\nabla^\beta \Delta[A^\alpha]) - 2\,\xi^\gamma\,(\nabla^\beta \Delta g1^{\alpha}{}_{\gamma}) - \\
&\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\beta \xi^\alpha) + 8\,\lambda\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\gamma A^\alpha) - 8\,\lambda\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\gamma A^\beta) + 2\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\gamma \xi^\alpha) - \\
&2\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\gamma \xi^\beta) + 8\,A^\gamma\,\Delta g1^{\beta}{}_{\delta}\,\xi_\gamma\,(\nabla^\delta A^\alpha) - 8\,A^\gamma\,\Delta g1^{\alpha}{}_{\delta}\,\xi_\gamma\,(\nabla^\delta A^\beta)\big)
\end{aligned}$$

Introducing the $\sqrt{-g}\ k^{\mu\nu}$

In[67]:=
```
δQϵ - 2 Antisymmetrize[Sqrt[-Detg[]] * Θ[α] ξ[β], {α, β}] // FullSimplification[];
k = (% /. RuleΘ) // FullSimplification[] // Factor
```

Out[68]=
$$\begin{aligned}
-\frac{1}{32\,G\,\pi}\sqrt{-\tilde{\tilde{g}}}\ \big(&4\,\lambda\,\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\alpha A^\beta) + 4\,A^\gamma\,\Delta g1^{\delta}{}_{\delta}\,\xi_\gamma\,(\nabla^\alpha A^\beta) + 8\,\Delta[A_\gamma]\,\xi^\gamma\,(\nabla^\alpha A^\beta) - \\
&8\,\Delta[A^\gamma]\,\xi^\beta\,(\nabla^\alpha A_\gamma) - 8\,\lambda\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\alpha A^\gamma) - 8\,A^\gamma\,\Delta g1^{\beta}{}_{\delta}\,\xi_\gamma\,(\nabla^\alpha A^\delta) + 8\,\lambda\,(\nabla^\alpha \Delta[A^\beta]) + \\
&8\,A^\gamma\,\xi_\gamma\,(\nabla^\alpha \Delta[A^\beta]) + 2\,\xi^\gamma\,(\nabla^\alpha \Delta g1^{\beta}{}_{\gamma}) - 2\,\xi^\beta\,(\nabla^\alpha \Delta g1^{\gamma}{}_{\gamma}) + \Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\alpha \xi^\beta) - \\
&4\,\lambda\,\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\beta A^\alpha) - 4\,A^\gamma\,\Delta g1^{\delta}{}_{\delta}\,\xi_\gamma\,(\nabla^\beta A^\alpha) - 8\,\Delta[A_\gamma]\,\xi^\gamma\,(\nabla^\beta A^\alpha) + \\
&8\,\Delta[A^\gamma]\,\xi^\alpha\,(\nabla^\beta A_\gamma) + 8\,\lambda\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\beta A^\gamma) + 8\,A^\gamma\,\Delta g1^{\alpha}{}_{\delta}\,\xi_\gamma\,(\nabla^\beta A^\delta) - \\
&8\,\lambda\,(\nabla^\beta \Delta[A^\alpha]) - 8\,A^\gamma\,\xi_\gamma\,(\nabla^\beta \Delta[A^\alpha]) - 2\,\xi^\gamma\,(\nabla^\beta \Delta g1^{\alpha}{}_{\gamma}) + 2\,\xi^\alpha\,(\nabla^\beta \Delta g1^{\gamma}{}_{\gamma}) - \\
&\Delta g1^{\gamma}{}_{\gamma}\,(\nabla^\beta \xi^\alpha) + 8\,\Delta[A^\gamma]\,\xi^\beta\,(\nabla_\gamma A^\alpha) - 8\,\Delta[A^\gamma]\,\xi^\alpha\,(\nabla_\gamma A^\beta) + 2\,\xi^\beta\,(\nabla_\gamma \Delta g1^{\alpha\gamma}) - \\
&2\,\xi^\alpha\,(\nabla_\gamma \Delta g1^{\beta\gamma}) + 8\,\lambda\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\gamma A^\alpha) - 8\,\lambda\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\gamma A^\beta) + 2\,\Delta g1^{\beta}{}_{\gamma}\,(\nabla^\gamma \xi^\alpha) - \\
&2\,\Delta g1^{\alpha}{}_{\gamma}\,(\nabla^\gamma \xi^\beta) + 8\,A^\gamma\,\Delta g1^{\beta}{}_{\delta}\,\xi_\gamma\,(\nabla^\delta A^\alpha) - 8\,A^\gamma\,\Delta g1^{\alpha}{}_{\delta}\,\xi_\gamma\,(\nabla^\delta A^\beta)\big)
\end{aligned}$$

The parametric variations $\hat{\delta}$ g :

Here, we introduce the parametric variations to be put into the $k^{\mu\nu}$. One can variate the dynamical fields $g_{\mu\nu}$ and A_μ with respect to all of the parameters m, a , and q. Nonetheless, using the linearity of the charges in the perturbations, and to speed up the calculations, we variate the dynamical fields with respect to each one of the parameters separately, and sum up the results eventually.

```
In[69]:= DefTensor[hatdelg$m[-α, -β], M, PrintAs -> "δ̂mg"]
DefTensor[hatdelA$m[-α], M, PrintAs -> "δ̂mA"]
DefTensor[hatdelg$a[-α, -β], M, PrintAs -> "δ̂ag"]
DefTensor[hatdelA$a[-α], M, PrintAs -> "δ̂aA"]
DefTensor[hatdelg$q[-α, -β], M, PrintAs -> "δ̂qg"]
DefTensor[hatdelA$q[-α], M, PrintAs -> "δ̂qA"]
DefConstantSymbol[δm, PrintAs → "δm"]
DefConstantSymbol[δa, PrintAs → "δa"]
DefConstantSymbol[δq, PrintAs → "δq"]
```

Variation with respect to the parameter m, i.e. $\hat{\delta}_m$ g and $\hat{\delta}_m$ A

```
In[78]:= MySimplify[g[-α, -β]];
(D[%, m] * δm) // Simplify;
MatrixForm[%]
```

Out[80]//MatrixForm=

$$\begin{pmatrix} \frac{2\,G\,\delta m\,r}{a^2\,Cos[\theta]^2+r^2} & 0 & 0 & -\frac{2\,a\,G\,\delta m\,r\,Sin[\theta]^2}{a^2\,Cos[\theta]^2+r^2} \\ 0 & \frac{2\,G\,\delta m\,r\,(a^2\,Cos[\theta]^2+r^2)}{(a^2+q^2+r\,(-2\,G\,m+r))^2} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\frac{2\,a\,G\,\delta m\,r\,Sin[\theta]^2}{a^2\,Cos[\theta]^2+r^2} & 0 & 0 & \frac{2\,a^2\,G\,\delta m\,r\,Sin[\theta]^4}{a^2\,Cos[\theta]^2+r^2} \end{pmatrix}$$

```
In[81]:= % // InputForm
```

Out[81]//InputForm=

```
{{(2*G*δm*r[])/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (-2*a*G*δm*r[]*Sin[θ[]]^2)/
   (a^2*Cos[θ[]]^2 + r[]^2)}, {0, (2*G*δm*r[]*(a^2*Cos[θ[]]^2 + r[]^2))/
   (a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0}, {0, 0, 0, 0},
 {(-2*a*G*δm*r[]*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0,
  (2*a^2*G*δm*r[]*Sin[θ[]]^4)/(a^2*Cos[θ[]]^2 + r[]^2)}}
```

```
In[82]:= AllComponentValues[hatdelg$m[{-α, -B}, {-β, -B}],
  {{(2 * G * δm * r[]) / (a ^ 2 * Cos[θ[]] ^ 2 + r[] ^ 2), 0, 0,
    (-2 * a * G * δm * r[] * Sin[θ[]] ^ 2) / (a ^ 2 * Cos[θ[]] ^ 2 + r[] ^ 2)},
   {0, (2 * G * δm * r[] * (a ^ 2 * Cos[θ[]] ^ 2 + r[] ^ 2)) /
     (a ^ 2 + q ^ 2 + r[] * (-2 * G * m + r[])) ^ 2, 0, 0}, {0, 0, 0, 0},
   {(-2 * a * G * δm * r[] * Sin[θ[]] ^ 2) / (a ^ 2 * Cos[θ[]] ^ 2 + r[] ^ 2), 0, 0,
    (2 * a ^ 2 * G * δm * r[] * Sin[θ[]] ^ 4) / (a ^ 2 * Cos[θ[]] ^ 2 + r[] ^ 2)}}];
ChangeComponents[hatdelg$m[{α, B}, {-β, -B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$m[{-α, -B}, {β, B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$m[{α, B}, {β, B}], hatdelg$m[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_m g^{\alpha}{}_{\beta} \to g^{\alpha\gamma}\ \hat{\delta}_m g_{\gamma\beta}$ in 0.199941 Seconds

Computed $\hat{\delta}_m g_{\alpha}{}^{\beta} \to g^{\beta\gamma}\ \hat{\delta}_m g_{\alpha\gamma}$ in 0.222454 Seconds

Computed $\hat{\delta}_m g_{\alpha}{}^{\beta} \to g^{\beta\gamma}\ \hat{\delta}_m g_{\alpha\gamma}$ in 0.201450 Seconds

Computed $\hat{\delta}_m g^{\alpha\beta} \to g^{\alpha\gamma}\ \hat{\delta}_m g_{\gamma}{}^{\beta}$ in 0.246624 Seconds

In[86]:=
```
MySimplify[A[-α]];
(D[%, m] * δm) // Simplify;
MatrixForm[%]
```

Out[88]//MatrixForm=

$$\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

In[89]:=
```
% // InputForm
```

Out[89]//InputForm=
```
{0, 0, 0, 0}
```

In[90]:=
```
AllComponentValues[hatdelA$m[{-α, -B}], {0, 0, 0, 0}];
ChangeComponents[hatdelA$m[{α, B}], hatdelA$m[{-ρ, -B}]];
```

Computed $\hat{\delta}_m A^{\alpha} \rightarrow g^{\alpha\beta}\ \hat{\delta}_m A_{\beta}$ in 0.045111 Seconds

In[92]:=
```
Ruleδg$m =
 MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$m[-α, -β]}, MetricOn → All]
RuleδA$m = MakeRule[{Perturbation[A[-α]], hatdelA$m[-α]}, MetricOn → All]
```

Out[92]= $\{\text{HoldPattern}[\Delta g^{1\underline{\alpha\beta}}] \mapsto \text{Module}[\{\},\ \hat{\delta}_m g^{\alpha\beta}]\}$

Out[93]= $\{\text{HoldPattern}[\Delta[A^{\underline{\alpha}}]] \mapsto \text{Module}[\{\},\ \hat{\delta}_m A^{\alpha}]\}$

Variation with respect to the parameter a, i.e. $\hat{\delta}_a$ g and $\hat{\delta}_a$ A

In[94]:=
```
MySimplify[g[-α, -β]];
(D[%, a] * δa) // Simplify;
MatrixForm[%]
```

Out[96]//MatrixForm=

$$\begin{pmatrix} \frac{2\, a\, \delta a \operatorname{Cos}[\theta]^2 (q^2 - 2 G m r)}{(a^2 \operatorname{Cos}[\theta]^2 + r^2)^2} & 0 & 0 & \frac{\delta a\ (\text{c}}{} \\ 0 & \frac{2\, a\, \delta a\, (-r^2 + \operatorname{Cos}[\theta]^2 (q^2 + r(-2 G m + r)))}{(a^2 + q^2 + r(-2 G m + r))^2} & 0 & \\ 0 & 0 & 2\, a\, \delta a \operatorname{Cos}[\theta]^2 & \\ \frac{\delta a\, (q^2 - 2 G m r)(-a^2 \operatorname{Cos}[\theta]^2 + r^2) \operatorname{Sin}[\theta]^2}{(a^2 \operatorname{Cos}[\theta]^2 + r^2)^2} & 0 & 0 & \frac{2\, a\, \delta a \operatorname{Sin}[\theta]^2 (a^4 \text{Co}}{} \end{pmatrix}$$

In[97]:=
```
% // InputForm
```

Out[97]//InputForm=
```
{{(2*a*δa*Cos[θ[]]^2*(q^2 - 2*G*m*r[]))/(a^2*Cos[θ[]]^2 + r[]^2)^2, 0, 0,
   (δa*(q^2 - 2*G*m*r[])*(-(a^2*Cos[θ[]]^2) + r[]^2)*Sin[θ[]]^2)/
    (a^2*Cos[θ[]]^2 + r[]^2)^2},
  {0, (2*a*δa*(-r[]^2 + Cos[θ[]]^2*(q^2 + r[]*(-2*G*m + r[]))))/
    (a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0}, {0, 0, 2*a*δa*Cos[θ[]]^2, 0},
  {(δa*(q^2 - 2*G*m*r[])*(-(a^2*Cos[θ[]]^2) + r[]^2)*Sin[θ[]]^2)/
    (a^2*Cos[θ[]]^2 + r[]^2)^2, 0, 0,
   (2*a*δa*Sin[θ[]]^2*(a^4*Cos[θ[]]^4 + 2*a^2*Cos[θ[]]^2*r[]^2 +
      r[]^2*(r[]^2 - (q^2 - 2*G*m*r[])*Sin[θ[]]^2)))/(a^2*Cos[θ[]]^2 + r[]^2)^2}}
```

In[98]:=
```
AllComponentValues[hatdelg$a[{-α, -B}, {-β, -B}],
  {{(2 * a * δa * Cos[θ[]]^2 * (q^2 - 2 * G * m * r[])) / (a^2 * Cos[θ[]]^2 + r[]^2)^2, 0,
    0, (δa * (q^2 - 2 * G * m * r[]) * (- (a^2 * Cos[θ[]]^2) + r[]^2) * Sin[θ[]]^2) /
     (a^2 * Cos[θ[]]^2 + r[]^2)^2},
   {0, (2 * a * δa * (-r[]^2 + Cos[θ[]]^2 * (q^2 + r[] * (-2 * G * m + r[])))) /
     (a^2 + q^2 + r[] * (-2 * G * m + r[]))^2, 0, 0}, {0, 0, 2 * a * δa * Cos[θ[]]^2, 0},
   {(δa * (q^2 - 2 * G * m * r[]) * (- (a^2 * Cos[θ[]]^2) + r[]^2) * Sin[θ[]]^2) /
     (a^2 * Cos[θ[]]^2 + r[]^2)^2, 0, 0,
    (2 * a * δa * Sin[θ[]]^2 * (a^4 * Cos[θ[]]^4 + 2 * a^2 * Cos[θ[]]^2 * r[]^2 +
         r[]^2 * (r[]^2 - (q^2 - 2 * G * m * r[]) * Sin[θ[]]^2))) /
     (a^2 * Cos[θ[]]^2 + r[]^2)^2}}];
ChangeComponents[hatdelg$a[{α, B}, {-β, -B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$a[{-α, -B}, {β, B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$a[{α, B}, {β, B}], hatdelg$a[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_a g^{\alpha}{}_{\beta} \to g^{\alpha\gamma}\ \hat{\delta}_a g_{\gamma\beta}$ in 0.212957 Seconds

Computed $\hat{\delta}_a g_{\alpha}{}^{\beta} \to g^{\beta\gamma}\ \hat{\delta}_a g_{\alpha\gamma}$ in 0.216880 Seconds

Computed $\hat{\delta}_a g_{\alpha}{}^{\beta} \to g^{\beta\gamma}\ \hat{\delta}_a g_{\alpha\gamma}$ in 0.199283 Seconds

Computed $\hat{\delta}_a g^{\alpha\beta} \to g^{\alpha\gamma}\ \hat{\delta}_a g_{\gamma}{}^{\beta}$ in 0.267155 Seconds

In[102]:=
```
MySimplify[A[-α]];
(D[%, a] * δa) // Simplify;
MatrixForm[%]
```

Out[104]//MatrixForm=

$$\begin{pmatrix} -\frac{2\, a\, q\, \delta a\, \mathrm{Cos}[\theta]^2\, r}{(a^2\, \mathrm{Cos}[\theta]^2+r^2)^2} \\ 0 \\ 0 \\ \frac{q\, \delta a\, r\, (a^2\, \mathrm{Cos}[\theta]^2-r^2)\, \mathrm{Sin}[\theta]^2}{(a^2\, \mathrm{Cos}[\theta]^2+r^2)^2} \end{pmatrix}$$

In[105]:=
```
% // InputForm
```

Out[105]//InputForm=
```
{(-2*a*q*δa*Cos[θ[]]^2*r[])/(a^2*Cos[θ[]]^2 + r[]^2)^2, 0, 0,
 (q*δa*r[]*(a^2*Cos[θ[]]^2 - r[]^2)*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2)^2}
```

In[106]:=
```
AllComponentValues[hatdelA$a[{-α, -B}],
  {(-2 * a * q * δa * Cos[θ[]]^2 * r[]) / (a^2 * Cos[θ[]]^2 + r[]^2)^2,
   0, 0, (q * δa * r[] * (a^2 * Cos[θ[]]^2 - r[]^2) * Sin[θ[]]^2) /
    (a^2 * Cos[θ[]]^2 + r[]^2)^2}];
ChangeComponents[hatdelA$a[{α, B}], hatdelA$a[{-ρ, -B}]];
```

Computed $\hat{\delta}_a A^{\alpha} \to g^{\alpha\beta}\ \hat{\delta}_a A_{\beta}$ in 0.045026 Seconds

In[108]:=
```
Ruleδg$a =
  MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$a[-α, -β]}, MetricOn → All]
RuleδA$a = MakeRule[{Perturbation[A[-α]], hatdelA$a[-α]}, MetricOn → All]
```

Out[108]= $\{\mathrm{HoldPattern}[\Delta g^{1}{}_{\underline{\alpha\beta}}] :\to \mathrm{Module}[\{\},\ \hat{\delta}_a g^{\alpha\beta}]\}$

Out[109]= $\{\mathrm{HoldPattern}[\Delta[A_{\underline{\alpha}}]] :\to \mathrm{Module}[\{\},\ \hat{\delta}_a A^{\alpha}]\}$

Variation with respect to the parameter q, i.e. $\hat{\delta}_q$ g and $\hat{\delta}_q$ A

In[110]:=
```
MySimplify[g[-α, -β]];
(D[%, q] * δq) // Simplify;
MatrixForm[%]
```

Out[112]//MatrixForm=

$$\begin{pmatrix} -\frac{2\,q\,\delta q}{a^2\,\text{Cos}[\theta]^2+r^2} & 0 & 0 & \frac{2\,a\,q\,\delta q\,\text{Sin}[\theta]^2}{a^2\,\text{Cos}[\theta]^2+r^2} \\ 0 & -\frac{2\,q\,\delta q\,(a^2\,\text{Cos}[\theta]^2+r^2)}{(a^2+q^2+r\,(-2\,G\,m+r))^2} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ \frac{2\,a\,q\,\delta q\,\text{Sin}[\theta]^2}{a^2\,\text{Cos}[\theta]^2+r^2} & 0 & 0 & -\frac{2\,a^2\,q\,\delta q\,\text{Sin}[\theta]^4}{a^2\,\text{Cos}[\theta]^2+r^2} \end{pmatrix}$$

In[113]:=
```
% // InputForm
```

Out[113]//InputForm=
```
{{(-2*q*δq)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0, (2*a*q*δq*Sin[θ[]]^2)/
   (a^2*Cos[θ[]]^2 + r[]^2)}, {0, (-2*q*δq*(a^2*Cos[θ[]]^2 + r[]^2))/
   (a^2 + q^2 + r[]*(-2*G*m + r[]))^2, 0, 0}, {0, 0, 0, 0},
 {(2*a*q*δq*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0,
  (-2*a^2*q*δq*Sin[θ[]]^4)/(a^2*Cos[θ[]]^2 + r[]^2)}}
```

In[114]:=
```
AllComponentValues[hatdelg$q[{-α, -B}, {-β, -B}],
  {{(-2 * q * δq) / (a^2 * Cos[θ[]]^2 + r[]^2), 0, 0,
    (2 * a * q * δq * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2)}, {0,
    (-2 * q * δq * (a^2 * Cos[θ[]]^2 + r[]^2)) / (a^2 + q^2 + r[] * (-2 * G * m + r[]))^2,
    0, 0}, {0, 0, 0, 0}, {(2 * a * q * δq * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2),
    0, 0, (-2 * a^2 * q * δq * Sin[θ[]]^4) / (a^2 * Cos[θ[]]^2 + r[]^2)}}];
ChangeComponents[hatdelg$q[{α, B}, {-β, -B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$q[{-α, -B}, {β, B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
ChangeComponents[hatdelg$q[{α, B}, {β, B}], hatdelg$q[{-ρ, -B}, {-β, -B}]];
```

Computed $\hat{\delta}_q\, g^{\alpha}{}_{\beta} \rightarrow g^{\alpha\gamma}\; \hat{\delta}_q\, g_{\gamma\beta}$ in 0.198283 Seconds

Computed $\hat{\delta}_q\, g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\; \hat{\delta}_q\, g_{\alpha\gamma}$ in 0.219834 Seconds

Computed $\hat{\delta}_q\, g_{\alpha}{}^{\beta} \rightarrow g^{\beta\gamma}\; \hat{\delta}_q\, g_{\alpha\gamma}$ in 0.247434 Seconds

Computed $\hat{\delta}_q\, g^{\alpha\beta} \rightarrow g^{\alpha\gamma}\; \hat{\delta}_q\, g_{\gamma}{}^{\beta}$ in 0.293439 Seconds

In[118]:=
```
MySimplify[A[-α]];
(D[%, q] * δq) // Simplify;
MatrixForm[%]
```

Out[120]//MatrixForm=

$$\begin{pmatrix} \frac{\delta q\,r}{a^2\,\text{Cos}[\theta]^2+r^2} \\ 0 \\ 0 \\ -\frac{a\,\delta q\,r\,\text{Sin}[\theta]^2}{a^2\,\text{Cos}[\theta]^2+r^2} \end{pmatrix}$$

In[121]:=
```
% // InputForm
```

Out[121]//InputForm=
```
{(δq*r[])/(a^2*Cos[θ[]]^2 + r[]^2), 0, 0,
 -((a*δq*r[]*Sin[θ[]]^2)/(a^2*Cos[θ[]]^2 + r[]^2))}
```

In[122]:=
```
AllComponentValues[hatdelA$q[{-α, -B}], {(δq * r[]) / (a^2 * Cos[θ[]]^2 + r[]^2),
    0, 0, -((a * δq * r[] * Sin[θ[]]^2) / (a^2 * Cos[θ[]]^2 + r[]^2))}];
ChangeComponents[hatdelA$q[{α, B}], hatdelA$q[{-ρ, -B}]];
```

Computed $\hat{\delta}_q A^{\alpha} \to g^{\alpha\beta}\ \hat{\delta}_q A_{\beta}$ in 0.043520 Seconds

In[124]:=
```
Ruleδg$q =
 MakeRule[{Perturbationg[LI[1], -α, -β], hatdelg$q[-α, -β]}, MetricOn → All]
RuleδA$q = MakeRule[{Perturbation[A[-α]], hatdelA$q[-α]}, MetricOn → All]
```

Out[124]= {HoldPattern[$\Delta g^{1\underline{\alpha\beta}}$] :→ Module[{}, $\hat{\delta}_q g^{\alpha\beta}$]}

Out[125]= {HoldPattern[Δ[$A^{\underline{\alpha}}$]] :→ Module[{}, $\hat{\delta}_q A^{\alpha}$]}

Mass

Defining the exact symmetry generator for the mass $\epsilon_M = \{\partial_t + \Omega_\infty \partial_\varphi,\ \Phi_\infty\}$:

Asymptotic angular velocity is equal to $\Omega_\infty = \frac{-g_{t\varphi}}{g_{\varphi\varphi}}\Big|_{r\to\infty}$

In[126]:=
```
DefConstantSymbol[Ω∞]
```

In[127]:=
```
MySimplify[g[-α, -β]];
Limit[-%[[1, 4]]/%[[4, 4]], r[] → ∞] // FullSimplify
```

Out[128]= 0

In[129]:=
```
RuleΩ∞ = MakeRule[{Ω∞, 0}, MetricOn → All]
```

Out[129]= {HoldPattern[Ω∞] :→ Module[{}, 0]}

In[130]:=
```
DefTensor[η[α], M]
AllComponentValues[η[{α, B}], {1, 0, 0, Ω∞}];
ChangeComponents[η[{-α, -B}], η[{ρ, B}]];
```

Computed $\eta_{\alpha} \to g_{\beta\alpha}\ \eta^{\beta}$ in 0.056177 Seconds

Asymptotic electric potentials $\Phi_\infty = \eta^{\mu} A_{\mu}\ |_{r\to\infty}$:

In[133]:=
```
DefConstantSymbol[Φ∞]
```

In[134]:=
```
Limit[MySimplify[η[-α] A[α]], r[] → ∞] /. RuleΩ∞ // FullSimplify
```

Out[134]= 0

In[135]:=
```
RuleΦ∞ = MakeRule[{Φ∞, 0}, MetricOn → All]
```

Out[135]= {HoldPattern[Φ∞] :→ Module[{}, 0]}

Rules to identify $\epsilon_M = \{\partial_t + \Omega_\infty \partial_\varphi , \Phi_\infty\}$:

```
In[136]:= Ruleη = MakeRule[{ξ[μ], η[μ]}, MetricOn → All]
          Ruleλ = MakeRule[{lambda[], Φ∞}, MetricOn → All]
```

```
Out[136]= {HoldPattern[ξ^μ] :→ Module[{}, η^μ]}

Out[137]= {HoldPattern[λ] :→ Module[{}, Φ∞]}
```

Calculating the variation of the mass with respect to the parameter m, i.e. the $\hat{\delta}_m M$

```
In[138]:= (k /. Ruleη /. Ruleλ /. Ruleδg$m /. RuleδA$m /. RuleΦ∞ /. RuleΩ∞) //
              ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
          % // ToBasis[B] // ToBasis[B] // ComponentArray;
```

One can use the pull-back of the Hodge dual of the result above to any closed codimension-2 surface surrounding the singularity at the origin. Nonetheless, pull-back to the surfaces of constant time and radius makes the calculations simpler. So, we choose the k^{01} component.

```
In[140]:= %[[1, 2]] // Factor // Simplify;
          % // MySimplify
```

Out[141]=
$$-\frac{1}{16\pi\,(a^2 + a^2 \mathrm{Cos}[2\theta] + 2 r^2)^2}\,\delta m\,\left(a^4 - a^5\,\Omega\infty - 12\,a^2\,r^2 + 5\,a^3\,\Omega\infty\,r^2 - 16\,r^4 + 12\,a\,\Omega\infty\,r^4 + a^3\,\mathrm{Cos}[4\theta]\,\left(-a + a^2\,\Omega\infty - \Omega\infty\,r^2\right) - 4\,a\,\mathrm{Cos}[2\theta]\,r^2\,\left(a + a^2\,\Omega\infty + 3\,\Omega\infty\,r^2\right)\right)\mathrm{Sin}[\theta]$$

One can integrate the result above on θ and φ, calculated on any arbitrary constant radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

```
In[142]:= Limit[%, r[] → ∞]
```

Out[142]=
$$\frac{\delta m\,(4 - 3\,a\,\Omega\infty + 3\,a\,\Omega\infty\,\mathrm{Cos}[2\theta])\,\mathrm{Sin}[\theta]}{16\pi}$$

```
In[143]:= Integrate[%, {θ[], 0, π}]
```

Out[143]=
$$\frac{\delta m - a\,\delta m\,\Omega\infty}{2\pi}$$

The $\hat{\delta}_m M$

```
In[144]:= Integrate[%, {φ[], 0, 2 π}]
```

Out[144]= $\delta m - a\,\delta m\,\Omega\infty$

```
In[145]:= δM$m = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```

Out[145]= δm

Calculating the variation of the mass with respect to the parameter a, i.e. the $\hat{\delta}_a M$

```
In[146]:= (k /. Ruleη /. Ruleλ /. Ruleδg$a /. RuleδA$a /. RuleΦ∞ /. RuleΩ∞) //
              ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
          % // ToBasis[B] // ToBasis[B] // ComponentArray;
```

Again, we choose the pull – back to the constant (t, r) surfaces,
i.e. the k^{01} component, for simplicity :

```
In[148]:= %[[1, 2]] // Factor // FullSimplify;
          % // MySimplify
```

Out[149]=
$$-\frac{1}{64\,G\,\pi\,(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2)^4}\,\delta a\,\Big(75\,a^7\,G\,m-5\,a^8\,G\,m\,\Omega\infty+12\,a^7\,G\,m\,\mathrm{Cos}[6\,\theta]+4\,a^8\,G\,m\,\Omega\infty\,\mathrm{Cos}[6\,\theta]+a^7\,G\,m\,\mathrm{Cos}[8\,\theta]+a^8\,G\,m\,\Omega\infty\,\mathrm{Cos}[8\,\theta]-75\,a^7\,r-104\,a^5\,q^2\,r+29\,a^6\,q^2\,\Omega\infty\,r-12\,a^7\,\mathrm{Cos}[6\,\theta]\,r+4\,a^5\,q^2\,\mathrm{Cos}[6\,\theta]\,r-16\,a^6\,q^2\,\Omega\infty\,\mathrm{Cos}[6\,\theta]\,r-a^7\,\mathrm{Cos}[8\,\theta]\,r-a^6\,q^2\,\Omega\infty\,\mathrm{Cos}[8\,\theta]\,r+456\,a^5\,G\,m\,r^2-61\,a^6\,G\,m\,\Omega\infty\,r^2+12\,a^5\,G\,m\,\mathrm{Cos}[6\,\theta]\,r^2+32\,a^6\,G\,m\,\Omega\infty\,\mathrm{Cos}[6\,\theta]\,r^2+a^6\,G\,m\,\Omega\infty\,\mathrm{Cos}[8\,\theta]\,r^2-248\,a^5\,r^3+480\,a^3\,q^2\,r^3-208\,a^4\,q^2\,\Omega\infty\,r^3-20\,a^5\,\mathrm{Cos}[6\,\theta]\,r^3-40\,a^4\,q^2\,\Omega\infty\,\mathrm{Cos}[6\,\theta]\,r^3+720\,a^3\,G\,m\,r^4-104\,a^4\,G\,m\,\Omega\infty\,r^4+28\,a^4\,G\,m\,\Omega\infty\,\mathrm{Cos}[6\,\theta]\,r^4-272\,a^3\,r^5+192\,a\,q^2\,r^5-48\,a^2\,q^2\,\Omega\infty\,r^5+192\,a\,G\,m\,r^6+144\,a^2\,G\,m\,\Omega\infty\,r^6-64\,a\,r^7+192\,G\,m\,\Omega\infty\,r^8+4\,a^2\,\mathrm{Cos}[4\,\theta]\,\big(a^6\,G\,m\,\Omega\infty+13\,a^5\,(G\,m-r)+26\,a^2\,\Omega\infty\,r^3\,(2\,q^2+G\,m\,r)+12\,\Omega\infty\,r^5\,(5\,q^2+G\,m\,r)+a^4\,\Omega\infty\,r\,(-7\,q^2+15\,G\,m\,r)-4\,a\,r^3\,(14\,q^2-11\,G\,m\,r+7\,r^2)-2\,a^3\,r\,(3\,q^2-23\,G\,m\,r+17\,r^2)\big)-4\,\mathrm{Cos}[2\,\theta]\,\big(a^8\,G\,m\,\Omega\infty-29\,a^7\,(G\,m-r)+48\,G\,m\,\Omega\infty\,r^8+48\,a^2\,\Omega\infty\,r^5\,(q^2+G\,m\,r)+4\,a^6\,\Omega\infty\,r\,(-q^2+2\,G\,m\,r)+a^4\,\Omega\infty\,r^3\,(-10\,q^2+7\,G\,m\,r)+48\,a\,r^5\,(-3\,q^2-3\,G\,m\,r+r^2)+32\,a^3\,r^3\,(-2\,q^2-7\,G\,m\,r+3\,r^2)+a^5\,r\,(33\,q^2-157\,G\,m\,r+91\,r^2)\big)\Big)\,\mathrm{Sin}[\theta]$$

One can integrate the result above on θ and φ, on any arbitrary radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

```
In[150]:= Limit[%, r[] → ∞]
```

Out[150]=
$$-\frac{3\,m\,\delta a\,\Omega\infty\,\mathrm{Sin}[\theta]^3}{8\,\pi}$$

```
In[151]:= Integrate[%, {θ[], 0, π}]
```

Out[151]=
$$-\frac{m\,\delta a\,\Omega\infty}{2\,\pi}$$

The $\hat{\delta}_a M$

```
In[152]:= Integrate[%, {φ[], 0, 2 π}]
```

Out[152]= $-m\,\delta a\,\Omega\infty$

```
In[153]:= δM$a = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```

Out[153]= 0

Calculating the variation of the mass with respect to the parameter q, using the $k^{\mu\nu}$

```
In[154]:= (k /. Ruleη /. Ruleλ /. Ruleδg$q /. RuleδA$q /. RuleΩ∞ /. RuleΦ∞) //
              ContractMetric // ToCanonical // FullSimplification[] // FullSimplify;
          % // ToBasis[B] // ToBasis[B] // ComponentArray;
```

Again, we choose the pull – back to the constant (t, r) surfaces,
i.e. the k^{01} component, for simplicity :

In[156]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify /. RuleΦ∞ /. RuleΩ∞ // Simplify
```

Out[157]=

$$-\frac{1}{16\, G\, \pi\, (a^2 + a^2\, \mathrm{Cos}[2\,\theta] + 2\, r^2)^3}\, a^2\, q\, \delta q\, r\, \left(36\, a^2 - 14\, a^3\, \Omega\infty + a^3\, \Omega\infty\, \mathrm{Cos}[6\,\theta] + 16\, r^2 - 4\, a\, \Omega\infty\, r^2 + 2\, a\, \mathrm{Cos}[4\,\theta]\left(-2\, a + 7\, a^2\, \Omega\infty + 10\, \Omega\infty\, r^2\right) + \mathrm{Cos}[2\,\theta]\left(32\, a^2 - a^3\, \Omega\infty + 48\, r^2 - 16\, a\, \Omega\infty\, r^2\right)\right) \mathrm{Sin}[\theta]$$

One can integrate the result above on θ and φ, on any arbitrary radius r>0. Nonetheless, the r→∞ makes the calculations simpler.

In[158]:=
```
Limit[%, r[] → ∞]
```

Out[158]= 0

In[159]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[159]= 0

The $\hat{\delta}_q M$

In[160]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[160]= 0

In[161]:=
```
δM$q = % /. RuleΦ∞ /. RuleΩ∞ // FullSimplify
```

Out[161]= 0

Now we can sum up all of the variations:

In[162]:=
```
δM = δM$m + δM$a + δM$q
```

Out[162]= δm

The result shows that δM is integrable. The integrated result is M=m. The constant of integration has been fixed by setting M=0 for the Minkowski spacttime.

Angular momentum

The exact symmetry generator for the angular momentum $\epsilon_J = \{-\partial_\varphi,\ 0\}$

In[163]:=
```
Undef[η]
DefTensor[η[α], M]
AllComponentValues[η[{α, B}], {0, 0, 0, -1}];
ChangeComponents[η[{-α, -B}], η[{ρ, B}]];
Ruleη = MakeRule[{ξ[μ], η[μ]}, MetricOn → All]
Ruleλ = MakeRule[{lambda[], 0}, MetricOn → All]
```

Computed $\eta_\alpha \to g_{\beta\alpha}\ \eta^\beta$ in 0.043553 Seconds

Out[167]= {HoldPattern[$\xi^{\underline{\mu}}$] :→ Module[{}, η^μ]}

Out[168]= {HoldPattern[λ] :→ Module[{}, 0]}

All the steps in calculating the variations of the angular momentum using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m J$

In[169]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$m /. RuleδA$m) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[172]= $-\left(\left(a\ \delta m\ \left(a^4-3\ a^2\ r^2-6\ r^4+a^2\ \mathrm{Cos}[2\ \theta]\ \left(a^2-r^2\right)\right)\ \mathrm{Sin}[\theta]^3\right)/\left(4\ \pi\ \left(a^2+a^2\ \mathrm{Cos}[2\ \theta]+2\ r^2\right)^2\right)\right)$

In[173]:=
```
Limit[%, r[] → ∞]
```

Out[173]= $\frac{3\ a\ \delta m\ \mathrm{Sin}[\theta]^3}{8\ \pi}$

In[174]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[174]= $\frac{a\ \delta m}{2\ \pi}$

In[175]:=
```
δJ$m = Integrate[%, {φ[], 0, 2 π}]
```

Out[175]= $a\ \delta m$

The $\hat{\delta}_a J$

In[176]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$a /. RuleδA$a) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[179]=

$$-\frac{1}{16\,G\,\pi\,(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2)^4}\,\delta a\,\Big(10\,a^8\,G\,m+a^8\,G\,m\,\mathrm{Cos}[6\,\theta]-46\,a^6\,q^2\,r-a^6\,q^2\,\mathrm{Cos}[6\,\theta]\,r+94\,a^6\,G\,m\,r^2+a^6\,G\,m\,\mathrm{Cos}[6\,\theta]\,r^2+168\,a^4\,q^2\,r^3+132\,a^4\,G\,m\,r^4+144\,a^2\,q^2\,r^5-48\,a^2\,G\,m\,r^6-96\,G\,m\,r^8+2\,a^4\,\mathrm{Cos}[4\,\theta]\,\big(3\,a^4\,G\,m+2\,r^3\,(-10\,q^2+7\,G\,m\,r)+a^2\,r\,(-9\,q^2+17\,G\,m\,r)\big)+a^2\,\mathrm{Cos}[2\,\theta]\,\big(15\,a^6\,G\,m+48\,r^5\,(5\,q^2+G\,m\,r)+32\,a^2\,r^3\,(4\,q^2+5\,G\,m\,r)+a^4\,r\,(-63\,q^2+127\,G\,m\,r)\big)\Big)\,\mathrm{Sin}[\theta]^3$$

In[180]:=
```
Limit[%, r[] → ∞]
```

Out[180]=

$$\frac{3\,m\,\delta a\,\mathrm{Sin}[\theta]^3}{8\,\pi}$$

In[181]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[181]=

$$\frac{m\,\delta a}{2\,\pi}$$

In[182]:=
```
δJ$a = Integrate[%, {φ[], 0, 2 π}]
```

Out[182]= $m\,\delta a$

The $\hat{\delta}_q J$

In[183]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$q /. RuleδA$q) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify // MySimplify
```

Out[185]=

$$-\frac{a^3\,q\,\delta q\,r\,(15\,a^2+a^2\,\mathrm{Cos}[4\,\theta]+12\,r^2+4\,\mathrm{Cos}[2\,\theta]\,(4\,a^2+5\,r^2))\,\mathrm{Sin}[\theta]^3}{4\,G\,\pi\,(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2)^3}$$

In[186]:=
```
Limit[%, r[] → ∞]
```

Out[186]= 0

In[187]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[187]= 0

In[188]:=
```
δJ$q = Integrate[%, {φ[], 0, 2 π}]
```

Out[188]= 0

Now we can sum up all of the variations:

In[189]:=
```
δJ = δJ$m + δJ$a + δJ$q
```

Out[189]= m δa + a δm

The result shows that δJ=δ(ma) total derivative and hence, it is integrable. The integrated result is J=ma in which the constant of integration has been fixed by the choice of J=0 for the Minkowski spacetime.

Electric Charge

The exact symmetry generator for the electric charge $\epsilon_Q = \{0, 1\}$

In[190]:=
```
Undef[η]
DefTensor[η[α], M]
AllComponentValues[η[{α, B}], {0, 0, 0, 0}];
ChangeComponents[η[{-α, -B}], η[{ρ, B}]];
Ruleη = MakeRule[{ξ[μ], η[μ]}, MetricOn → All]
Ruleλ = MakeRule[{lambda[], 1}, MetricOn → All]
```

Computed $\eta_\alpha \to g_{\beta\alpha}\ \eta^\beta$ in 0.039911 Seconds

Out[194]= {HoldPattern[$\xi^{\underline{\mu}}$] :→ Module[{}, η^μ]}

Out[195]= {HoldPattern[λ] :→ Module[{}, 1]}

All the steps in calculating the variations of the electric charge using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m Q$

In[196]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$m /. RuleδA$m) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[199]= 0

In[200]:=
```
Limit[%, r[] → ∞]
```

Out[200]= 0

In[201]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[201]= 0

In[202]:=
```
δQ$m = Integrate[%, {φ[], 0, 2 π}]
```

Out[202]= 0

The $\hat{\delta}_a Q$

In[203]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$a /. RuleδA$a) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[206]= $-\frac{a\, q\, \delta a\, r^2 \left(9 a^2 - a^2 \mathrm{Cos}[4\theta] + 4 r^2 + 4 \mathrm{Cos}[2\theta] \left(2 a^2 + 3 r^2\right)\right) \mathrm{Sin}[\theta]}{2 G \pi \left(a^2 + a^2 \mathrm{Cos}[2\theta] + 2 r^2\right)^3}$

In[207]:=
```
Limit[%, r[] → ∞]
```

Out[207]= 0

In[208]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[208]= 0

In[209]:=
```
δQ$a = Integrate[%, {φ[], 0, 2 π}]
```

Out[209]= 0

The $\hat{\delta}_q Q$

In[210]:=
```
(k /. Ruleη /. Ruleλ /. Ruleδg$q /. RuleδA$q) // ContractMetric // ToCanonical //
    FullSimplification[] // FullSimplify;
% // ToBasis[B] // ToBasis[B] // ComponentArray;
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[213]= $-\frac{\delta q \left(a^2 + a^2 \mathrm{Cos}[2\theta] - 2 r^2\right) \left(a^2 + r^2\right) \mathrm{Sin}[\theta]}{2 G \pi \left(a^2 + a^2 \mathrm{Cos}[2\theta] + 2 r^2\right)^2}$

In[214]:=
```
Limit[%, r[] → ∞]
```

Out[214]= $\frac{\delta q\, \mathrm{Sin}[\theta]}{4 G \pi}$

In[215]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[215]= $\frac{\delta q}{2 G \pi}$

In[216]:=
```
δQ$q = Integrate[%, {φ[], 0, 2 π}]
```

Out[216]= $\frac{\delta q}{G}$

Now we can sum up all of the variations:

In[217]:=
```
δQ = δQ$m + δQ$a + δQ$q
```

Out[217]= $\frac{\delta q}{G}$

The result shows that $\delta Q = \delta\left(\frac{q}{G}\right)$ is a total derivative and hence, it is integrable. The integrated result is $Q = \frac{q}{G}$ in which the constant of integration has been fixed by the choice of Q=0 for the Minkowski spacetime.

Entropy

BIn order to identify the exact symmetry generator for the entropy, we need to calculate some entities associated to the horizon :

Horizon radius r_H :

In[218]:=
```
Solve[Δ == 0 /. r[] → r, r]
```

Out[218]= $\left\{\left\{r \to G\,m - \sqrt{-a^2 + G^2 m^2 - q^2}\right\}, \left\{r \to G\,m + \sqrt{-a^2 + G^2 m^2 - q^2}\right\}\right\}$

In[219]:=
```
rH = G m + Sqrt[-a^2 + G^2 m^2 - q^2];
```

Horizon angular velocity $\Omega_H = \frac{-g_{t\varphi}}{g_{\varphi\varphi}}\Big|_{r\to r_H}$:

In[220]:=
```
DefConstantSymbol[ΩH]
```

In[221]:=
```
MySimplify[g[-α, -β]];
Limit[(-1 * %[[1, 4]])/%[[4, 4]], r[] → rH] // FullSimplify
```

Out[222]= $\frac{a}{-q^2 + 2\,G\,m\left(G\,m + \sqrt{-a^2 + G^2 m^2 - q^2}\right)}$

In[223]:=
```
RuleΩH = MakeRule[{ΩH, a / (-q^2 + 2 G m (G m + Sqrt[-a^2 + G^2 m^2 - q^2]))}]
```

Out[223]= $\left\{\text{HoldPattern}[\Omega H] \mapsto \text{Module}\left[\{\}, \frac{a}{-q^2 + 2\,G\,m\left(G\,m + \sqrt{-a^2 + G^2 m^2 - q^2}\right)}\right]\right\}$

Horizon Killing vector $\eta_H = \partial_t + \Omega_H\,\partial_\varphi$:

In[224]:=
```
DefTensor[ηH[α], M]
AllComponentValues[ηH[{α, B}], {1, 0, 0, ΩH}];
ChangeComponents[ηH[{-α, -B}], ηH[{ρ, B}]];
```

Computed $\eta H_\alpha \to g_{\beta\alpha}\ \eta H^\beta$ in 0.047845 Seconds

Horizon electri c potentials $\Phi_H = \eta_H{}^\mu A_\mu\ |_{r\to r_H}$:

In[227]:=
```
DefConstantSymbol[ΦH]
```

In[228]:=
```
(ηH[-α] A[α] // MySimplify) /. r[] → rH /. RuleΩH // FullSimplify
```
Out[228]=
$$\frac{2\,a^2\,G\,m\,q+q^3\left(G\,m-\sqrt{-a^2+G^2\,m^2-q^2}\right)}{4\,a^2\,G^2\,m^2+q^4}$$

In[229]:=
```
RuleΦH = MakeRule[
  {ΦH, (2 a^2 G m q + q^3 (G m - Sqrt[-a^2 + G^2 m^2 - q^2])) / (4 a^2 G^2 m^2 + q^4)}, MetricOn → All]
```
Out[229]=
$$\left\{\text{HoldPattern}[\Phi H] :\to \text{Module}\left[\{\},\ \frac{2\,a^2\,G\,m\,q}{4\,a^2\,G^2\,m^2+q^4}+\frac{G\,m\,q^3}{4\,a^2\,G^2\,m^2+q^4}-\frac{q^3\,\sqrt{-a^2+G^2\,m^2-q^2}}{4\,a^2\,G^2\,m^2+q^4}\right]\right\}$$

Finding the Hawking temperature T_H:

Finding the surface gravity κH on the horizon

In[230]:=
```
-1/2 (CD[-μ][ηH[-ν]]) * (CD[μ][ηH[ν]]);
% // MySimplify;
Sqrt[%] /. r[] → rH /. θ[] → π/2 /. RuleΩH // Simplify // Expand // FullSimplify
```
Out[232]=
$$\sqrt{\left(\frac{1}{\left(4\,a^2\,G^2\,m^2+q^4\right)^2}\left(a^2-G^2\,m^2+q^2\right)\left(4\,a^2\,G^2\,m^2-8\,G^4\,m^4+8\,G^2\,m^2\,q^2-q^4+8\,G^3\,m^3\,\sqrt{-a^2+G^2\,m^2-q^2}-4\,G\,m\,q^2\,\sqrt{-a^2+G^2\,m^2-q^2}\right)\right)}$$

To make the result simpler, we multiply it by $1=\frac{(rH^2+a^2)}{(rH^2+a^2)}$

In[233]:=
```
κH = (% * Sqrt[(rH^2 + a^2)^2] // ExpandNumerator // FullSimplify) / (rH^2 + a^2)
```
Out[233]=
$$\frac{\sqrt{-a^2+G^2\,m^2-q^2}}{a^2+\left(G\,m+\sqrt{-a^2+G^2\,m^2-q^2}\right)^2}$$

Hawking temperature of the horizon, $TH=\frac{\kappa H}{2\pi}$

In[234]:=
```
DefConstantSymbol[TH]
```
In[235]:=
```
RuleTH = MakeRule[{TH, κH/(2 π)}];
```

Defining the exact symmetry for the entropy $\epsilon_H=\frac{1}{T_H}\{\partial_t+\Omega_H\,\partial_\varphi\,,\ -\Phi_H\}$

In[236]:=
```
RuleηH = MakeRule[{ξ[μ], 1/TH ηH[μ]}, MetricOn → All];
RuleλH = MakeRule[{lambda[], -ΦH/TH}, MetricOn → All];
```

All the steps in calculating the variations of the entropy using the $k^{\mu\nu}$ are the same as the calculations of the mass. So, we will not repeat the descriptions provided above.

The $\hat{\delta}_m S$

In[238]:=
```
(k /. RuleηH /. RuleλH /. Ruleδg$m /. RuleδA$m) // Factor // ContractMetric //
  ToCanonical // FullSimplification[] // FullSimplify
```

Out[238]=

$$\frac{1}{32\,G\,\pi\,TH}\sqrt{-\tilde{g}}\,\Big(2\,\eta H^{\gamma}\big(-(\nabla^{\alpha}\hat{\delta}_m g^{\beta}{}_{\gamma}) + \nabla^{\beta}\hat{\delta}_m g^{\alpha}{}_{\gamma}\big) + \hat{\delta}_m g^{\gamma}{}_{\gamma}\big(4\,\Phi H(\nabla^{\alpha}A^{\beta}) - \nabla^{\alpha}\eta H^{\beta} - 4\,\Phi H(\nabla^{\beta}A^{\alpha}) + \nabla^{\beta}\eta H^{\alpha}\big) + 2\Big(4\,\Phi H\big(\nabla^{\alpha}\hat{\delta}_m A^{\beta} - \nabla^{\beta}\hat{\delta}_m A^{\alpha} + \hat{\delta}_m g^{\alpha\gamma}(\nabla^{\beta}A_{\gamma} - \nabla_{\gamma}A^{\beta})\big) + 4\,\hat{\delta}_m A^{\gamma}\big(\eta H_{\gamma}(-(\nabla^{\alpha}A^{\beta}) + \nabla^{\beta}A^{\alpha}) + \eta H^{\beta}(\nabla^{\alpha}A_{\gamma} - \nabla_{\gamma}A^{\alpha}) + \eta H^{\alpha}(-(\nabla^{\beta}A_{\gamma}) + \nabla_{\gamma}A^{\beta})\big) + \eta H^{\beta}(\nabla^{\alpha}\hat{\delta}_m g^{\gamma}{}_{\gamma} - \nabla_{\gamma}\hat{\delta}_m g^{\alpha\gamma}) + \eta H^{\alpha}(-(\nabla^{\beta}\hat{\delta}_m g^{\gamma}{}_{\gamma}) + \nabla_{\gamma}\hat{\delta}_m g^{\beta\gamma}) - \hat{\delta}_m g^{\beta\gamma}(4\,\Phi H(\nabla^{\alpha}A_{\gamma}) - 4\,\Phi H(\nabla_{\gamma}A^{\alpha}) + \nabla_{\gamma}\eta H^{\alpha}) + \hat{\delta}_m g^{\alpha\gamma}(\nabla_{\gamma}\eta H^{\beta}) + 2\,A^{\gamma}\,\eta H_{\gamma}\big(\hat{\delta}_m g^{\delta}{}_{\delta}(-(\nabla^{\alpha}A^{\beta}) + \nabla^{\beta}A^{\alpha}) + 2(-(\nabla^{\alpha}\hat{\delta}_m A^{\beta}) + \nabla^{\beta}\hat{\delta}_m A^{\alpha} + \hat{\delta}_m g^{\beta\delta}(\nabla^{\alpha}A_{\delta} - \nabla_{\delta}A^{\alpha}) + \hat{\delta}_m g^{\alpha\delta}(-(\nabla^{\beta}A_{\delta}) + \nabla_{\delta}A^{\beta}))\big)\Big)\Big)$$

In[239]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[240]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[241]=

$$-\left(\left(\delta m\left(a^4 - a^5\,\Omega H - 12\,a^2 r^2 + 5\,a^3\,\Omega H\,r^2 - 16\,r^4 + 12\,a\,\Omega H\,r^4 + a^3\,\mathrm{Cos}[4\,\theta]\left(-a + a^2\,\Omega H - \Omega H\,r^2\right) - 4\,a\,\mathrm{Cos}[2\,\theta]\,r^2\left(a + a^2\,\Omega H + 3\,\Omega H\,r^2\right)\right)\mathrm{Sin}[\theta]\right) \Big/ \left(16\,\pi\,TH\left(a^2 + a^2\,\mathrm{Cos}[2\,\theta] + 2\,r^2\right)^2\right)\right)$$

In[242]:=
```
Limit[%, r[] → ∞]
```

Out[242]=

$$\frac{\delta m\,(4 - 3\,a\,\Omega H + 3\,a\,\Omega H\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta]}{16\,\pi\,TH}$$

In[243]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[243]=

$$\frac{\delta m - a\,\delta m\,\Omega H}{2\,\pi\,TH}$$

In[244]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[244]=

$$\frac{\delta m - a\,\delta m\,\Omega H}{TH}$$

In[245]:=
```
δS$m = % /. RuleΩH /. RuleTH // FullSimplify
```

Out[245]=

$$-\frac{2\,\pi\left(a^2 + q^2 - 2\,G\,m\left(G\,m + \sqrt{-a^2 + G^2 m^2 - q^2}\right)\right)\delta m}{\sqrt{-a^2 + G^2 m^2 - q^2}}$$

The $\hat{\delta}_a S$

In[246]:=
```
(k /. RuleηH /. RuleλH /. Ruleδg$a /. RuleδA$a) // Factor // ContractMetric //
    ToCanonical // FullSimplification[] // FullSimplify
```

Out[246]=

$$\frac{1}{32\, G\, \pi\, TH}\sqrt{-\tilde{\tilde{g}}}\,\Big(2\,\eta H^{\gamma}\left(-\left(\nabla^{\alpha}\hat{\delta}_a g^{\beta}{}_{\gamma}\right)+\nabla^{\beta}\hat{\delta}_a g^{\alpha}{}_{\gamma}\right)+\hat{\delta}_a g^{\gamma}{}_{\gamma}\left(4\,\Phi H\left(\nabla^{\alpha}A^{\beta}\right)-\nabla^{\alpha}\eta H^{\beta}-4\,\Phi H\left(\nabla^{\beta}A^{\alpha}\right)+\nabla^{\beta}\eta H^{\alpha}\right)+2\,\Big(4\,\Phi H\left(\nabla^{\alpha}\hat{\delta}_a A^{\beta}-\nabla^{\beta}\hat{\delta}_a A^{\alpha}+\hat{\delta}_a g^{\alpha\gamma}\left(\nabla^{\beta}A_{\gamma}-\nabla_{\gamma}A^{\beta}\right)\right)+4\,\hat{\delta}_a A^{\gamma}\left(\eta H_{\gamma}\left(-\left(\nabla^{\alpha}A^{\beta}\right)+\nabla^{\beta}A^{\alpha}\right)+\eta H^{\beta}\left(\nabla^{\alpha}A_{\gamma}-\nabla_{\gamma}A^{\alpha}\right)+\eta H^{\alpha}\left(-\left(\nabla^{\beta}A_{\gamma}\right)+\nabla_{\gamma}A^{\beta}\right)\right)+\eta H^{\beta}\left(\nabla^{\alpha}\hat{\delta}_a g^{\gamma}{}_{\gamma}-\nabla_{\gamma}\hat{\delta}_a g^{\alpha\gamma}\right)+\eta H^{\alpha}\left(-\left(\nabla^{\beta}\hat{\delta}_a g^{\gamma}{}_{\gamma}\right)+\nabla_{\gamma}\hat{\delta}_a g^{\beta\gamma}\right)-\hat{\delta}_a g^{\beta\gamma}\left(4\,\Phi H\left(\nabla^{\alpha}A_{\gamma}\right)-4\,\Phi H\left(\nabla_{\gamma}A^{\alpha}\right)+\nabla_{\gamma}\eta H^{\alpha}\right)+\hat{\delta}_a g^{\alpha\gamma}\left(\nabla_{\gamma}\eta H^{\beta}\right)+2\,A^{\gamma}\,\eta H_{\gamma}\Big(\hat{\delta}_a g^{\delta}{}_{\delta}\left(-\left(\nabla^{\alpha}A^{\beta}\right)+\nabla^{\beta}A^{\alpha}\right)+2\left(-\left(\nabla^{\alpha}\hat{\delta}_a A^{\beta}\right)+\nabla^{\beta}\hat{\delta}_a A^{\alpha}+\hat{\delta}_a g^{\beta\delta}\left(\nabla^{\alpha}A_{\delta}-\nabla_{\delta}A^{\alpha}\right)+\hat{\delta}_a g^{\alpha\delta}\left(-\left(\nabla^{\beta}A_{\delta}\right)+\nabla_{\delta}A^{\beta}\right)\right)\Big)\Big)\Big)$$

In[247]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[248]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[249]=

$$\Big(\delta a\,\Big(-75\,a^7\,G\,m+5\,a^8\,G\,m\,\Omega H-12\,a^7\,G\,m\,\mathrm{Cos}[6\,\theta]-4\,a^8\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]-a^7\,G\,m\,\mathrm{Cos}[8\,\theta]-a^8\,G\,m\,\Omega H\,\mathrm{Cos}[8\,\theta]+75\,a^7\,r+104\,a^5\,q^2\,r-29\,a^6\,q^2\,\Omega H\,r+12\,a^7\,\mathrm{Cos}[6\,\theta]\,r-4\,a^5\,q^2\,\mathrm{Cos}[6\,\theta]\,r+16\,a^6\,q^2\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r+a^7\,\mathrm{Cos}[8\,\theta]\,r+a^6\,q^2\,\Omega H\,\mathrm{Cos}[8\,\theta]\,r-456\,a^5\,G\,m\,r^2+416\,a^5\,q\,\Phi H\,r^2+61\,a^6\,G\,m\,\Omega H\,r^2-12\,a^5\,G\,m\,\mathrm{Cos}[6\,\theta]\,r^2-16\,a^5\,q\,\Phi H\,\mathrm{Cos}[6\,\theta]\,r^2-32\,a^6\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^2-a^6\,G\,m\,\Omega H\,\mathrm{Cos}[8\,\theta]\,r^2+248\,a^5\,r^3-480\,a^3\,q^2\,r^3+208\,a^4\,q^2\,\Omega H\,r^3+20\,a^5\,\mathrm{Cos}[6\,\theta]\,r^3+40\,a^4\,q^2\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^3-720\,a^3\,G\,m\,r^4+896\,a^3\,q\,\Phi H\,r^4+104\,a^4\,G\,m\,\Omega H\,r^4-28\,a^4\,G\,m\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r^4+272\,a^3\,r^5-192\,a\,q^2\,r^5+48\,a^2\,q^2\,\Omega H\,r^5-192\,a\,G\,m\,r^6+256\,a\,q\,\Phi H\,r^6-144\,a^2\,G\,m\,\Omega H\,r^6+64\,a\,r^7-192\,G\,m\,\Omega H\,r^8-4\,a^2\,\mathrm{Cos}[4\,\theta]\,\Big(a^6\,G\,m\,\Omega H+13\,a^5\,(G\,m-r)+26\,a^2\,\Omega H\,r^3\left(2\,q^2+G\,m\,r\right)+12\,\Omega H\,r^5\left(5\,q^2+G\,m\,r\right)+a^4\,\Omega H\,r\left(-7\,q^2+15\,G\,m\,r\right)-4\,a\,r^3\left(14\,q^2-11\,G\,m\,r+8\,q\,\Phi H\,r+7\,r^2\right)-2\,a^3\,r\left(3\,q^2-23\,G\,m\,r+12\,q\,\Phi H\,r+17\,r^2\right)\Big)+4\,\mathrm{Cos}[2\,\theta]\,\Big(a^8\,G\,m\,\Omega H-29\,a^7\,(G\,m-r)+48\,G\,m\,\Omega H\,r^8+48\,a^2\,\Omega H\,r^5\left(q^2+G\,m\,r\right)+4\,a^6\,\Omega H\,r\left(-q^2+2\,G\,m\,r\right)+a^4\,\Omega H\,r^3\left(-10\,q^2+7\,G\,m\,r\right)+48\,a\,r^5\left(-3\,q^2-3\,G\,m\,r+4\,q\,\Phi H\,r+r^2\right)+32\,a^3\,r^3\left(-2\,q^2-7\,G\,m\,r+8\,q\,\Phi H\,r+3\,r^2\right)+a^5\,r\left(33\,q^2-157\,G\,m\,r+132\,q\,\Phi H\,r+91\,r^2\right)\Big)\Big)\,\mathrm{Sin}[\theta]\Big)\Big/\left(64\,G\,\pi\,TH\left(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2\right)^4\right)$$

In[250]:=
```
Limit[%, r[] → ∞]
```

Out[250]=

$$-\frac{3\,m\,\delta a\,\Omega H\,\mathrm{Sin}[\theta]^3}{8\,\pi\,TH}$$

In[251]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[251]=

$$-\frac{m\,\delta a\,\Omega H}{2\,\pi\,TH}$$

In[252]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[252]=

$$-\frac{m\,\delta a\,\Omega H}{TH}$$

In[253]:=
```
δS$a = % /. RuleΩH /. RuleTH // FullSimplify
```

Out[253]=
$$-\frac{2\,a\,m\,\pi\,\delta a}{\sqrt{-a^2+G^2\,m^2-q^2}}$$

The $\hat{\delta}_q\,S$

In[254]:=
```
(k /. RuleηH /. RuleλH /. Ruleδg$q /. RuleδA$q) // Factor // ContractMetric //
    ToCanonical // FullSimplification[] // FullSimplify
```

Out[254]=
$$\begin{aligned}
&\frac{1}{32\,G\,\pi\,TH}\sqrt{-\tilde{\tilde{g}}}\,\Big(2\,\eta H^{\gamma}\,\big(-(\nabla^{\alpha}\hat{\delta}_q g^{\beta}{}_{\gamma})+\nabla^{\beta}\hat{\delta}_q g^{\alpha}{}_{\gamma}\big)+\\
&\quad\hat{\delta}_q g^{\gamma}{}_{\gamma}\,\big(4\,\Phi H\,(\nabla^{\alpha}A^{\beta})-\nabla^{\alpha}\eta H^{\beta}-4\,\Phi H\,(\nabla^{\beta}A^{\alpha})+\nabla^{\beta}\eta H^{\alpha}\big)+\\
&\quad 2\,\Big(4\,\Phi H\,\big(\nabla^{\alpha}\hat{\delta}_q A^{\beta}-\nabla^{\beta}\hat{\delta}_q A^{\alpha}+\hat{\delta}_q g^{\alpha\gamma}\,(\nabla^{\beta}A_{\gamma}-\nabla_{\gamma}A^{\beta})\big)+\\
&\qquad 4\,\hat{\delta}_q A^{\gamma}\,\big(\eta H_{\gamma}\,(-(\nabla^{\alpha}A^{\beta})+\nabla^{\beta}A^{\alpha})+\eta H^{\beta}\,(\nabla^{\alpha}A_{\gamma}-\nabla_{\gamma}A^{\alpha})+\eta H^{\alpha}\,(-(\nabla^{\beta}A_{\gamma})+\nabla_{\gamma}A^{\beta})\big)+\\
&\qquad \eta H^{\beta}\,\big(\nabla^{\alpha}\hat{\delta}_q g^{\gamma}{}_{\gamma}-\nabla_{\gamma}\hat{\delta}_q g^{\alpha\gamma}\big)+\eta H^{\alpha}\,\big(-(\nabla^{\beta}\hat{\delta}_q g^{\gamma}{}_{\gamma})+\nabla_{\gamma}\hat{\delta}_q g^{\beta\gamma}\big)-\\
&\qquad \hat{\delta}_q g^{\beta\gamma}\,\big(4\,\Phi H\,(\nabla^{\alpha}A_{\gamma})-4\,\Phi H\,(\nabla_{\gamma}A^{\alpha})+\nabla_{\gamma}\eta H^{\alpha}\big)+\hat{\delta}_q g^{\alpha\gamma}\,(\nabla_{\gamma}\eta H^{\beta})+\\
&\qquad 2\,A^{\gamma}\,\eta H_{\gamma}\,\Big(\hat{\delta}_q g^{\delta}{}_{\delta}\,(-(\nabla^{\alpha}A^{\beta})+\nabla^{\beta}A^{\alpha})+2\,\big(-(\nabla^{\alpha}\hat{\delta}_q A^{\beta})+\nabla^{\beta}\hat{\delta}_q A^{\alpha}+\\
&\qquad\qquad \hat{\delta}_q g^{\beta\delta}\,(\nabla^{\alpha}A_{\delta}-\nabla_{\delta}A^{\alpha})+\hat{\delta}_q g^{\alpha\delta}\,(-(\nabla^{\beta}A_{\delta})+\nabla_{\delta}A^{\beta})\big)\Big)\Big)\Big)
\end{aligned}$$

In[255]:=
```
% // ToBasis[B] // ToBasis[B] // ComponentArray;
```

In[256]:=
```
%[[1, 2]] // Factor // Simplify;
% // MySimplify
```

Out[257]=
$$\begin{aligned}
&\Big(\delta q\,\big(12\,a^6\,\Phi H-36\,a^4\,q\,r+14\,a^5\,q\,\Omega H\,r-a^5\,q\,\Omega H\,\mathrm{Cos}[6\,\theta]\,r+\\
&\quad 12\,a^4\,\Phi H\,r^2-16\,a^2\,q\,r^3+4\,a^3\,q\,\Omega H\,r^3-32\,a^2\,\Phi H\,r^4-32\,\Phi H\,r^6+\\
&\quad a^2\,\mathrm{Cos}[2\,\theta]\,\big(16\,a^4\,\Phi H+a^3\,q\,\Omega H\,r-48\,q\,r^3+16\,a\,q\,\Omega H\,r^3+16\,a^2\,r\,(-2\,q+\Phi H\,r)\big)+\\
&\quad 2\,a^3\,\mathrm{Cos}[4\,\theta]\,\big(2\,a^3\,\Phi H-7\,a^2\,q\,\Omega H\,r-10\,q\,\Omega H\,r^3+2\,a\,r\,(q+\Phi H\,r)\big)\big)\\
&\quad \mathrm{Sin}[\theta]\Big)\Big/\Big(16\,G\,\pi\,TH\,\big(a^2+a^2\,\mathrm{Cos}[2\,\theta]+2\,r^2\big)^3\Big)
\end{aligned}$$

In[258]:=
```
Limit[%, r[] → ∞]
```

Out[258]=
$$-\frac{\delta q\,\Phi H\,\mathrm{Sin}[\theta]}{4\,G\,\pi\,TH}$$

In[259]:=
```
Integrate[%, {θ[], 0, π}]
```

Out[259]=
$$-\frac{\delta q\,\Phi H}{2\,G\,\pi\,TH}$$

In[260]:=
```
Integrate[%, {φ[], 0, 2 π}]
```

Out[260]=
$$-\frac{\delta q\,\Phi H}{G\,TH}$$

In[261]:=
```
δS$q = % /. RuleΩH /. RuleTH /. RuleΦH // FullSimplify
```

Out[261]= $-\frac{2\,\pi\,q\left(1+\frac{G\,m}{\sqrt{-a^2+G^2\,m^2-q^2}}\right)\delta q}{G}$

Now we can sum up all of the variations:

In[262]:=
```
δS = δS$m + δS$a + δS$q
```

Out[262]= $-\frac{2\,a\,m\,\pi\,\delta a}{\sqrt{-a^2+G^2\,m^2-q^2}}-\frac{2\,\pi\left(a^2+q^2-2\,G\,m\left(G\,m+\sqrt{-a^2+G^2\,m^2-q^2}\right)\right)\delta m}{\sqrt{-a^2+G^2\,m^2-q^2}}-\frac{2\,\pi\,q\left(1+\frac{G\,m}{\sqrt{-a^2+G^2\,m^2-q^2}}\right)\delta q}{G}$

The result shows that $\delta S=\delta\left(\frac{4\,\pi(rH^2+a^2)}{4\,G}\right)$ is a total derivative and hence, it is integrable. To check this claim:

In[263]:=
```
(δS - ((D[4 π (rH^2 + a^2) / (4 G), m] * δm) + (D[4 π (rH^2 + a^2) / (4 G), a] * δa) +
      (D[4 π (rH^2 + a^2) / (4 G), q] * δq))) // FullSimplify
```

Out[263]= 0

The integrated result is $S=\frac{4\,\pi(rH^2+a^2)}{4\,G}$ in which the constant of integration has been fixed by the choice of S=0 for the Minkowski spacetime.

First law of thermodynamics

In the "solution phase space method," the first law originates from the local identity between the generators of entropy and other charges:
$\epsilon_H=\frac{1}{T_H}(\epsilon_M-(\Omega_H-\Omega_\infty)\,\epsilon_J-(\Phi_H-\Phi_\infty)\,\epsilon_Q)$. By the linearity of charge variations in their generators, one easily proves the first law as:
$\delta S_H=\frac{1}{T_H}(\delta M-(\Omega_H-\Omega_\infty)\,\delta J-(\Phi_H-\Phi_\infty)\,\delta Q)$. To cross check:

In[264]:=
```
(δS - 1/TH (δM - (ΩH - Ω∞) δJ - (ΦH - Φ∞) δQ)) /. RuleΦH /. RuleΦ∞ /. RuleΩH /. RuleΩ∞ /.
  RuleTH // FullSimplify
```

Out[264]= 0

For a review on the "solution phase space method, the papers below can be refered to:
1) K. Hajian, Gen.Rel.Grav. 48 (2016) no.8, 114, arXiv:1602.05575 [gr-qc].

2) M. Ghodrati, K. Hajian, M.R. Setare, arXiv:1606.04353 [hep-th].